

A listening bot for IRC

Judith: The code font is still difficult to see on the printed page. Can we get it darker??
-rls

Judith: I decreased the font size of @T so the whole title would fit. If it's too small, just use "Whispers." Thnx. -rls

Whispers from Beyond

A listening bot on an IRC channel wakes up when it hears certain keywords and notifies a defined user via instant messaging. *By Mike Schilli*

Open source projects like Catalyst use IRC channels to provide support; experts wait for user requests and then step in to give help. That said, IRC chat can make it difficult for the helpers to focus on their ongoing work. And, if the channel is full, conversations are always in full swing. The Perl bot I will describe in this article listens on a specific IRC channel and notifies its master when certain keywords occur.

The first step in creating an IRC bot, is fairly simple. After all, the CPAN Bot::BasicBot module that I've covered before provides an easily extensible framework for any kind of IRC bot. But how can the bot attract the attention of its hard-working user? Instant messaging with pop-up dialogs is one useful approach, and Pidgin provides a versatile client that supports common protocols such as Yahoo! Messenger, Google Talk, AIM, or MSN.

Chat via Web API

Some time ago, Yahoo! opened a web API [1] to its Messenger service whereby users would first log in and then use HTTP requests to exchange messages with other Yahoo! Messenger users. The bot script introduced here, `irc2ym`, joins an IRC channel and then just shuts up and listens (Figure 1). If a chat user mentions one of the keywords (Figure 2) in the `~/irc-keywords` file, the bot launches the `ymsend` script, which logs into the Messenger Web API and sends

the eavesdropped text message to a predefined Messenger account (Figure 3). The Messenger service then notifies the user, who immediately interrupts work, turns to the IRC channel, and contributes expert knowledge to help hapless newbies find their way around.

Sniffing Messages

Listing 1 [2] derives a `YMBot` class from the `Bot::BasicBot` base class on CPAN and overloads its `said()` method, which the bot calls whenever a user says something on an IRC channel. Along with a reference to the object, the bot passes a hash data structure to the method, containing the username in the `who` field and the message text in `body`.

In this callback method, the bot then calls the `keyword_match()` function, defined in line 58, and the function compares the message text with a dictionary of keywords parsed from the `~/irc2ym-keywords` file (Figure 4). The script parses the entries in the file and stores them in the global `@KEYWORD_LIST` array. If one of the regu-

lar expressions stored in the `@KEYWORD_LIST` array fits the bill, line 27 of the same file triggers the `ymsend` script in the same directory. This script accepts the message text at the command line, logs in to the Web API, performs a couple of authorization steps based on the OAuth protocol, and finally sends the message text to the user defined in `$recipient` in line 11 of Listing 2.

The script needs to jump through a few authentication hoops first, requiring the name of the sending Messenger user, their password, an API key that you need to retrieve from the Yahoo! Developer Network [3], and a shared secret for the application.

OAuth Jungle

The OAuth protocol [4] [5] lets an authenticated user pass a token to an application, which then acts on behalf of the user for a certain period of time. The beauty of the concept is that users don't have to tell the third-party application their password directly. The protocol authenticates just like other online offers at Yahoo's login screen, which then issues the token for the application to use. The concept makes a lot of sense with web applications, because users get trained never to enter their credentials on third-party sites, but



MIKE SCHILLI

Mike: I was contacted at mschilli@perlmeister.com. Mike's homepage can be found at <http://perlmeister.com>.




Figure 1: The ymbot does nothing until somebody mentions one of the predefined keywords.




Figure 2: An IRC participant named "hubbelquadrat" mentions the "cpan" keyword, and the eavesdropping bot notifies the user.



Figure 3: The bot has forwarded the message to the Y! Messenger user.

only on the original login screen of the provider. This issue is less evident with desktop applications like my script, which need the user's password anyway to authenticate at the login site behind the scenes.

In the case of Y! Messenger, the token allows the application (i.e., the script) to send messages to the IM network and receive responses for one hour. Because

the script runs very rarely and immediately quits after sending the message, storing the token wouldn't offer significant advantages. Thus, the script re-authenticates against the Yahoo login page, passing in a username and password (hard coded as \$user and \$password in ymsend) with every run, then picks up a new access token and uses it to run the send command in the web API.

In line 45, the ymsend script logs in the user as \$user and \$passwd at the URL stored in \$login_url. Yahoo! sends back a request token in the body of the response.

The script then sends the token and the API key, with a matching secret key, secret, to the next URL, \$auth_token_url, which then generates an access token, oauth_token, and an oauth_token_secret. The web server response uses

LISTING 1: irc2ym

```

01 #!/usr/local/bin/perl -w
02 use strict;
03 use local::lib;
04
05 #####
06 package YMBot;
07 #####
08 use base qw( Bot::BasicBot );
09 use FindBin qw($Bin);
10
11 my $ymsend = "$Bin/ymsend";
12 my ($home) = glob "~";
13 my $KEYWORD_LIST_FILE =
14   "$home/.irc2ym-keywords";
15 my @KEYWORD_LIST = ();
16
17 keyword_list_read();
18
19 #####
20 sub said {
21   #####
22   my ($self, $data) = @_;
23
24   if ( keyword_match(
25     $data->{body} ) ) {
26
27     my $rc = system( $ymsend,
28       "$data->{who} said: " .
29       "'$data->{body}'" );
30
31     warn "$ymsend failed: $!"
32       if $rc;
33   }
34
35   return $data;
36 }
37
38 #####
39 sub keyword_list_read {
40   #####
41   if (!open FILE,
42     "<$KEYWORD_LIST_FILE" ) {
43     warn "$KEYWORD_LIST_FILE ",
44       "not found";
45     return;
46   }
47
48   while (<FILE>) {
49     chomp;
50     s/#.*//;
51     next if /^\\s*$/;
52     push @KEYWORD_LIST, $_;
53   }
54   close FILE;
55 }
56
57 #####
58 sub keyword_match {
59   #####
60   my ($said) = @_;
61
62   for
63     my $regex (@KEYWORD_LIST)
64   {
65     return 1
66       if $said =~ /$regex/i;
67   }
68   return 0;
69 }
70
71 #####
72 package main;
73 #####
74 use Bot::BasicBot;
75
76 my $bot = YMBot->new(
77   server =>
78     "irc.freenode.com",
79   channels => ["#ymtest"],
80   nick => "ymbot",
81   name => "Relay to Y!M",
82   charset => "utf-8",
83 );
84
85 $bot->run();

```




Figure 4: The list of keywords to which the IRC bot will react.

the format `field=value&field=value ...`, which the script simply stores in a URI object in line 86 as a made-up query part of the URL. It then tells the `query_form` method to parse the object – this works because the data are formatted exactly like a URL using query parameters.

The combination of token and secret identifies the application as authorized by the user to use the web service on his behalf. The script then passes these on to the Messenger web service using the `$session_url`, which starts a new Messenger session and logs in the `$user` into the Yahoo! Messenger network. Once the session has started, other IM users see the user appear in their buddy lists, and the script uses the POST method in lines 148-155 to send the message passed in at the command line to the Messenger user defined in `$recipient` (who should be logged in). This last step involves encoding the request in JSON format as in:

```
{ message : "the message" }
```

If the message text also contains quotes, these non-standard characters must be

encoded correctly. The `qqquote` function exported by the `Sysadm::Install` CPAN module makes light work of this task.

Creating the Auth Token

To create an authentication token with a secret for the newly created application (i.e., the `ymsend` script), the API developer must click through *My Projects* and *New Project* (Yahoo! account required) on the Yahoo! Developer Network [3].

These steps will take you to the pop-up box shown in Figure 5. Because this is not a web application running in a browser, but a desktop client, you'll need to select *Or an application using these APIs: BOSS, Contacts, Mail, ...*

In the form that appears, the developer must enter a short name (e.g., `irc2ymessenger`) and a couple of words of explanation as the description (Figure 6). The *Kind of Application* drop-down box must be set to *Client/Desktop* (not *Web-based*).

Below *Access Scopes*, you can then select *This app requires access to private user data*, then in the mass of sub-items

that appears, just select the *Read/Write* option below the entry for *Yahoo! Messenger* (see Figure 7).

After accepting the conditions of use, you'll be given the keys you need to put the messenger client together (Figure 8). Cut and paste these into the strings in lines 15 and 16 of the `ymsend` script to set the `$api_key` and `$secret` variables.

In line 10 of the script, you'll also need to enter the password for the Messenger account sending the message. The username in the example is `zangzongzing`. If you don't have an account yet, you can simply press




Figure 7: The application requires read/write access to Yahoo! Messenger data.




Figure 8: The ready-made API keys for creating the Y! Messenger client.

the *Sign Up* link to let yahoo.com take you to the account registration page.

After this, you only need to create a list of keywords in `~/irc-keywords` and launch the `irc2ym` bot. The bot could take up to 20 seconds to log in to the preset channel on a heavily used IRC server, but then the bot will appear in the online list as `ymbot`.

Popular IRC clients include Irssi (for the command line), or Pidgin, the jack of all trades, which will display an ongoing chat once you are logged into the IRC channel.

If a channel participant uses one of the predefined keywords, `ymsend` will wake up and use the Messenger protocol to send the message to the predefined (and hopefully logged in) IM user, `$recipient`, in a dialog window. Now, it's time to help the newbies! ■■■

INFO

- [1] Yahoo! Messenger IM API: <http://developer.yahoo.com/messenger/guide/ch02.html>
- [2] Listings for this article: <http://www.linux-magazine.com/Resources/Article-Code>
- [3] Yahoo! Developer Network: <http://developer.yahoo.com/>
- [4] Documentation for authentication token: <http://developer.yahoo.com/messenger/guide/chapterintrotomessengersdk.html>
- [5] OAuth: <http://en.wikipedia.org/wiki/Oauth>

Judith: These figs are OK. -rls

Mike: irc2ym-keywords?? -rls

Judith: Figure 6 should be token-create-form. -rls

Judith: This is Figure 5. (Use the Fig. 5 caption above.) -rls




Figure 5: Developers need to request a consumer key for a desktop client application.




Figure 6: The developer must request an authentication token for a desktop client application.

LISTING 2: ymsend

```

001 #!/usr/local/bin/perl -w
002 use strict;
003 use LWP::UserAgent;
004 use Sysadm::Install
005   qw(qquote);
006 use URI;
007 use JSON;
008
009 my $user = "zangzongzang";
010 my $passwd = "*****";
011 my $recipient =
012   "mikeschilli";
013
014 my $api_key =
015   "*****";
016 my $secret = "*****";
017
018 my $login_url =
019   "https://login.yahoo.com/WSLogin/
020   v1/get_auth_token";
021 my $auth_token_url =
022   "https://api.login.yahoo.com/oauth/
023   v2/get_token";
024 my $session_url =
025   "http://developer.messenger.
026   yahooapis.com/v1/session";
027 my $message_url =
028   "http://developer.messenger.
029   yahooapis.com/v1/message/
030   yahoo/$recipient";
031
032 my ($msg) = join ' ', @ARGV;
033
034 die "usage: $0 message"
035   unless length $msg;
036
037 my $ua =
038   LWP::UserAgent->new();
039
040 my $url =
041   URI->new($login_url);
042
043 $url->query_form(
044   login => $user,
045   passwd => $passwd,
046   oauth_consumer_key =>
047     $api_key
048 );
049
050 my $resp = $ua->get($url);
051
052 if ($resp->is_error()) {
053   die
054     "Can't get request token: ",
055     $resp->message(), " ",
056     $resp->content();
057 }
058
059 my $u = URI->new($auth_token_url);
060
061 $u->query_form(
062   oauth_consumer_key =>
063     $api_key,
064   oauth_nonce =>
065     int(rand 10000000),
066   oauth_signature =>
067     "$secret&",
068   oauth_signature_method =>
069     "PLAINTEXT",
070   oauth_timestamp => time(),
071   oauth_token =>
072     $request_token,
073   oauth_version => "1.0"
074 );
075
076 $resp = $ua->get($u);
077
078 if ($resp->is_error()) {
079   die
080     "Can't get access token: ",
081     $resp->message(), " ",
082     $resp->content();
083 }
084
085 my $u = URI->new();
086 $u->query($resp->content());
087 my %form = $u->query_form;
088
089 my $session_url =
090   URI->new($session_url);
091
092 $session_url->query_form(
093   oauth_consumer_key =>
094     $api_key,
095   oauth_nonce =>
096     int(rand 10000000),
097   oauth_signature =>
098     "$secret&" .
099     $form{oauth_token_secret},
100   oauth_signature_method =>
101     "PLAINTEXT",
102   oauth_timestamp => time(),
103   oauth_token =>
104     $form{oauth_token},
105   oauth_version => "1.0"
106 );
107
108 $resp = $ua->post(
109   $session_url,
110   Content_Type =>
111     "application/json; " .
112     "charset=utf-8",
113   Content =>
114     q[
115       { "presenceState" : 0,
116         "presenceMessage" : "I'm
117           alive!"
118       }
119     ];
120
121 if ($resp->is_error()) {
122   die "Can't get session: ",
123     $resp->message(), " ",
124     $resp->content();
125 }
126
127 my $data = from_json(
128   $resp->content());
129
130 my $message_url =
131   URI->new($message_url);
132
133 $message_url->query_form(
134   oauth_consumer_key =>
135     $api_key,
136   oauth_nonce =>
137     int(rand 10000000),
138   oauth_signature =>
139     "$secret&" .
140     $form{oauth_token_secret},
141   oauth_signature_method =>
142     "PLAINTEXT",
143   oauth_timestamp => time(),
144   oauth_token =>
145     $form{oauth_token},
146   oauth_version => "1.0",
147   sid => $data->{sessionId},
148 );
149
150 $resp = $ua->post(
151   $message_url,
152   Content_Type =>
153     "application/json; " .
154     "charset=utf-8",
155   Content => '{ "message" : '
156     . qquote($msg) . ' }'
157 );
158
159 if ($resp->is_error()) {
160   die "Can't send message: ",
161     $resp->message(), " ",
162     $resp->content();
163 }

```